
Hierarchical Design

This lecture describes Verilog modules and parameters.
After this lecture you should be able to: declare modules with parameters and ports, and instantiate modules using posi-
tional, named and wildcard parameters and signals.

Modules

Simple things are easier to design and test than com-
plex ones. Thus it’s good practice to divide designs
into smaller parts1. These can often be re-used.
Many designs incorporate complex parts designed

by others (e.g. processors, memories and interfaces),
called design IP (“Intellectual Property”).
In Verilog each part is a module. Modules describe

logic that can be “instantiated” (duplicated and in-
serted into) another module:

instantiating module

instantiated module

signal
port

name:id

The module’s interfaces are defined by a module
statement describing ports and parameters. Ports are
in, out or inout (bidirectional) signals while param-
eters are values that can customize each instance of
a module. The module’s body contains additional
signal declarations and parallel (concurrently exe-
cuting) statements between module and endmodule.
These define the structure or behaviour of the mod-
ule.
Here’s an example of a module named bits that

defines an nb-bit register:
module bits
#(parameter nb=1)
(
input logic [nb-1:0] d,
output logic [nb-1:0] q,
input logic clock
) ;

always_ff @(posedge clock) q <= d ;

endmodule

1How small? A good rule of thumb is to make sure each part
can be described on a single page.

The parameter nb has a default value of 1 which
is used if a value is not specified when this module is
instantiated. There are two input ports (named d and
clock) and one output port (named q).
Amodule instantiation starts with the name of the

module followed by parameter values (if any), an in-
stance name (to identify individual instances of the
same module), and a description of how to connect
signals in the instantiating module to the ports in the
instantiated module. For example, the statement:
bits #(4) b0 (a,b,c) ;

would instantiate a bits module giving the first pa-
rameter a value 4, giving this instance the name b0,
and connect the signals a, b and c in the instantiat-
ing module to the corresponding ports in a copy (an
instantiation) of the bits module (in this case, d, q
and clock respectively).
Exercise 1: Draw a diagram for this instantiation of the bitsmod‑
ule. Label the module, instance, signal and port names as in the
diagram above.
An 8-bit, 3-stage shift register could be built using

three bitsmodules:
sr3bytes

newest oldest
bits:b0

d q

clock

d q

clock

d q

clock

clock

a b
bits:b1 bits:b2

module sr3bytes
(
input logic [7:0] newest,
output logic [7:0] oldest,
input logic clock
) ;

localparam nbits = 8 ;

logic [nbits-1:0] a, b ;

// matching by order
bits #(nbits) b0 (newest,a,clock);

// matching by name (order does not matter)

Figure 1: Shift Register Synthesis

bits #(.nb(nbits)) b1 (.q(b),.clock,.d(a));

// wildcards for names that match
bits #(.nb(nbits)) b2 (.d(b),.q(oldest),.*);

endmodule

Exercise2: Identify themodule instantiation statements in thecode
above. For each one, what is the instantiated module’s name? The
instance name?
When one module is instantiated in another, a sig-

nal can be connected to a module port by:

• port order (signal),

• port name and explicit signal name
(.port(signal)),

• port name only – connecting to the matching
signal name (.port),

• a wildcard that matches all remainingmatching
port and signal names (.*).

The signal name can be an expressions (e.g.
word[15:8]) instead of a signal. Port order and
port name syntax cannot both be used in the same
instantiation.
Matching of values to parameters can be done by

order (value) or explicitly, (.parameter(value)) –
but not both.
Note that we declared the variables a and bwithin

the instantiating module, sr3bytes. We need to use
signal names in module instantiations. This means
that any signal used in an instantiation of a module
must also be declared within the instantiating mod-
ule, possibly as an input or output of the instantiating
module.
The synthesis result, shown above, is as expected.

Exercise 3: Write a counter module with a width parameter w,
w‑bit inputs and outputs d and q, enable, load and down control

inputs, and a clock clk. The default counter width should be 4 bits,
the default value of enable should be 1, and the default values of
load, down and d should be 0. Rewrite the traffic light controller
module by instantiating two instances of this module.

